# $Id: PKGBUILD 68300 2012-03-21 10:35:13Z pschmitz $
# Maintainer: Ronald van Haren <ronald.archlinux.org>
# Contributor: JJDaNiMoTh <jjdanimoth@gmail.com>
# Contributor: nesl247 <nesl247@gmail.com>

pkgname=compiz-fusion-plugins-main
pkgver=0.8.8
pkgrel=2
pkgdesc="Compiz Fusion Main plugins"
arch=('i686' 'x86_64')
url="http://www.compiz.org"
license=('GPL')
depends=('compiz-core' 'libjpeg>=7' 'librsvg' 'compiz-bcop' 'libxdamage' 'libxcomposite'
	'libxinerama' 'startup-notification')
makedepends=('intltool' 'pkgconfig' 'gettext')
groups=('compiz-fusion' 'compiz-fusion-kde' 'compiz-fusion-gtk')
conflicts=('compiz-fusion-plugins-main-git')
source=(http://releases.compiz-fusion.org/${pkgver}/compiz-plugins-main-${pkgver}.tar.bz2)
options=(!libtool)
sha1sums=('3c99841b744ae6c4a3edf2019626484880607dc6')

build() {
  cd "${srcdir}/compiz-plugins-main-${pkgver}"

  CPPFLAGS="$CPPFLAGS -I/usr/include/compiz -I/usr/include/startup-notification-1.0 \
        -I/usr/include/libxml2" ./configure --prefix=/usr --sysconfdir=/etc

  make
}

package() {
  cd "${srcdir}/compiz-plugins-main-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
